

Analóg és digitális áramkörök tesztelése és programozása peremfigyeléssel – 5.

Dr. Kohut József főiskolai docens, Molnár Zsolt főiskolai tanársegéd

Budapesti Műszaki Főiskola Kandó Kálmán Villamosmérnöki Kar Műszertechnikai és Automatizálási Intézet

A peremfigyeléses tesztelés eddig számba vett és tárgyalt lehetőségei, és kedvező tulajdonságai mellett mindenképpen meg kell említeni az automatikus tesztminta-generálás és teszteredmény-feldolgozás lehetőségét is. Ahhoz, hogy ezeket a lehetőségeket ki lehessen használni, feltétlenül szükséges egy olyan leíró állomány, amely jellemzi a peremfigyeléses alkatrészeket.

A **lábkiosztás** megadásával a logikai neveket fizikai lábakkal rendeljük össze. Lehetőség van többféle, előre definiált tokozásához is lábkiosztást megadni. (A **PIN_MAP_STRING** sorai logikailag egyetlen karaktersort alkotnak, az egyes sorok végén lévő „&” karakter fűzi össze őket.) Példa:

```
attribute PIN_MAP of XYZ_IC :
    entity is PHYSICAL_PIN_MAP;
constant DW : PIN_MAP_STRING:= "SEL:(1,13), " &
    "A:(2,3,4,5), Y: 6, GND: 7, VCC:14, TDI: 8, " &
    "TDO: 9, TCK: 10, TMS: 11, NC: 12";
```

A **TAP definiálásánál** adható meg, hogy a TAP jelei közül melyek vannak megvalósítva az adott eszközben, illetve TCK-n a maximálisan megengedett órajel-frekvencia. Példánkban a TRST-jelel nincs megvalósítva a TAP-ban, a maximális órajel 40 MHz. Az órajel akár a magas, akár az alacsony szintű állapotában leállítható („BOTH” jelzi).

```
attribute TAP_SCAN_IN of TDI : signal is true;
attribute TAP_SCAN_MODE of TMS : signal is true;
attribute TAP_SCAN_OUT of TDO : signal is true;
attribute TAP_SCAN_CLOCK of TCK :
    signal is (40.0e6, BOTH);
```

Az **utasításregiszter leírásában** megtalálhatók a vele kapcsolatos, eszközfüggő jellemzők. Ezek: az utasításregiszter hossza, a megvalósított utasítások kódja, a TAP-vezérlő egy bizonyos állapotában (Capture-IR, utasításregiszter feltöltése) az utasítás beléptetésével egyidejűleg kilépő (ügynevezett capture) bitminta, valamint az utasítások közül azok megjelölése, amelyeket nem nyilvános használatra szántak. Példánkban az utasításregiszter hossza 4 bit, és az eszköz nem támogat a kötelező utasításokon kívül más utasításokat.

```
attribute INSTRUCTION_LENGTH of XYZ_IC:
    entity is 4;
attribute INSTRUCTION_OPCODE of XYZ_IC: entity is
    "BYPASS (1111), " &
    "EXTEST (0000), " &
    "SAMPLE (1010)"
attribute INSTRUCTION_CAPTURE of XYZ_IC:
    entity is "1001";
```

Az **adatútvonalba kapcsolható regiszterek megadásával** rögzíti a gyártó, hogy az egyes adatregiszterek melyik utasítás végrehajtása során kapcsolódnak a TDI- és TDO-lábak közé. Példánkban az EXTEST- és a SAMPLE-parancsok esetén a peremfigyelő regiszter (**BOUNDARY**), a BYPASS-parancs esetén pedig az áthidaló regiszter (**BYPASS**) kapcsolódik a TDI- és TDO-lábak közé.

```
attribute REGISTER_ACCESS of XYZ_IC : entity is
    "BOUNDARY (EXTEST, SAMPLE), " &
    "BYPASS (BYPASS)";
```


A **peremfigyelő regiszter leírása** egy peremfigyelőcella-listát tartalmaz, amelyből kiolvasható többek között a peremfigyelő celláknak a logikai jelekhez való hozzárendelése, a cellák kapcsolódási sorrendje és vezérlése. A cellák számozása nullától kezdődik, a nulladik peremfigyelő cella van legközelebb a TDO kivezetéshez. Példa:

```
attribute BOUNDARY_LENGTH of XYZ_IC: entity is 7;
attribute BOUNDARY_REGISTER of XYZ_IC : entity is
    „0 (BC_1, SEL(0), input, X),” &
    „1 (BC_1, SEL(1), input, X),” &
    „2 (BC_1, A(0), input, X),” &
    „3 (BC_1, A(1), input, X),” &
    „4 (BC_1, A(2), input, X),” &
    „5 (BC_1, A(3), input, X),” &
    „6 (BC_1, Y, output3, X),”
```

A következőkben egy konkrét típus, a Texas Instruments által gyártott SN74ABT8245 8-szoros, kétirányú, peremfigyelési képességgel rendelkező buszmeghajtón keresztül mutatjuk be a BSDL-t. A fenti típus az SN74LS245, azonos funkciójú eszköz IEEE1149.1-kompatibilis változata. Az 1. ábrán látható, hogy a két áramkör tokozása csupán a négy TAP-kivezetésben tér el.

Az SN74ABT8245 gyártó által kibocsátott BSDL fájlját (terjedelmi okok miatt a megjegyzések egy részének törlése után) a 2...6. ábráinkon részletekben mutatjuk be, illetve felhívjuk a figyelmet egyes sajátos megoldásokra a teljesség igénye nélkül. Az ismertetésben csak azokkal a sorokkal foglalkozunk, amelyek az eddigiek alapján nem egyértelműek. Az „eredeti” BSDL-forrást keretezéssel jelöltük.

1. ábra A 74LS245 és a 74ABT8245 tokozása



```

-----
-- TI SN74ABT8245                                     --
-- IEEE Std 1149.1 (JTAG) Boundary-Scan Test Device --
-- with Octal Bus Transceivers                       --
-----

```

2. ábra

A 2. ábrán látható programsorok a gépi feldolgozás számára nem adnak információt, csak a BSDL-fájlt olvasó személyt tájékoztatják. A „--” (két mínuszjel) karakterekkel kezdődő sorok megjegyzések.

```

entity sn74abt8245 is
    generic (PHYSICAL_PIN_MAP : string := "UNDEFINED");
    port (DIR:in bit;
          A:inout bit_vector(1 to 8);
          B:inout bit_vector(1 to 8);
          OE_NEG:in bit;
          GND,VCC:linkage bit;
          TDO:out bit;
          TDI,TMS,TCK:in bit;
          NC:linkage bit_vector(1 to 4));
    use STD_1149_1_1990.all;
    attribute PIN_MAP of sn74abt8245 : entity is PHYSICAL_PIN_MAP;
    constant JT : PIN_MAP_STRING := "DIR:1, R:(23,22,21," &
        "20,19,17,16,15), B:(2,3,4,5,7,8,9,10)," &
        "OE_NEG:24, GND:6, VCC:18, TDO:11, TMS:12," &
        "TCK:13, TDI:14";
    constant BW : PIN_MAP_STRING := "DIR:1, R:(23,22,21," &
        "20,19,17,16,15), B:(2,3,4,5,7,8,9,10)," &
        "OE_NEG:24, GND:6, VCC:18, TDO:11, TMS:12," &
        "TCK:13, TDI:14";
    constant FK : PIN_MAP_STRING := "DIR:9, R:(6,5,4,3,2," &
        "27,26,25), B:(10,11,12,13,16,17,18,19)," &
        "GND:14, VCC:28, TDO:20, TMS:21, TCK:23," &
        "OE_NEG:7, TDI:24, NC:(1,8,15,22)";
    attribute TAP_SCAN_IN of TDI : signal is true;
    attribute TAP_SCAN_MODE of TMS : signal is true;
    attribute TAP_SCAN_OUT of TDO : signal is true;
    attribute TAP_SCAN_CLOCK of TCK : signal is (50.0e6, BOTH);
    attribute INSTRUCTION_LENGTH of sn74abt8245 : entity is 8;
    attribute INSTRUCTION_OPCODE of sn74abt8245 : entity is
        "BYPASS (1111111, 10000001)," &
        "EXTEST (00000000)," &
        "SAMPLE (10000010)," &
        "INTEST (00000011)," &
        "HIGHZ (00000110)," & -- Control Boundary to High-Impedance
        "CLAMP (10000111)," & -- Control Boundary to 1/0
        "RUNT (00001001)," & -- Boundary Run Test
        "READBN (00001010)," & -- Boundary Read Normal Mode
        "READBT (10001011)," & -- Boundary Read Test Mode
        "CELLST(00001100)," & -- Boundary Self-Test Normal Mode
        "TOPHIP (10001101)," & -- Boundary Toggle Outputs Test Mode
        "SCANCN (10001110)," & -- BCR Scan Normal Mode
        "SCANCT (00001111)" ; -- BCR Scan Test Mode
    attribute INSTRUCTION_CAPTURE of sn74abt8245 : entity is "10000001";
    attribute INSTRUCTION_DISABLE of sn74abt8245 : entity is "HIGHZ";
    attribute INSTRUCTION_GUREAD of sn74abt8245 : entity is "CLAMP";

```

3. ábra

A 3. ábrán látható programrészlet sorai az IEEE1149.1 kötelező és opcionális utasításain kívül számos utasítást definiálnak (például RUNT, READBN). Az utasítások egy része speciális szerepet kap. Például a HIGHZ-utasítás az alapfunkcióján kívül (min-

```

attribute REGISTER_ACCESS of sn74abt8245 : entity is
    "BOUNDARY (EXTEST, SAMPLE, INTEST, READBN, READBT, CELLST
    "BYPASS (BYPASS, HIGHZ, CLAMP, RUNT, TOPHIP)," &
    "BCR[11] (SCANCN, SCANCT)" ;

```

4. ábra

den peremfigyeléses láb nagyimpedanciás állapotba helyezése) a TDI- és TDO pontokat összekötő a BYPASS regiszteren keresztül, azaz az adott eszköznek nem lehet parancsot adni.

A 4. ábra programrészletében két utasítás esetén (SCANCN és SCANCT) az alkatrészgyártó által definiált és megvalósított 11 bites regiszter, a BCR regiszter kapcsolódik a TDI és a TDO lábak közé.

```

attribute BOUNDARY_CELLS of sn74abt8245 : entity is "BC_1";
attribute BOUNDARY_LENGTH of sn74abt8245 : entity is 36;
attribute BOUNDARY_REGISTER of sn74abt8245 : entity is

"0 (BC_1, R(1) ,output3 , X, 34, 0, Z)," &
"1 (BC_1, R(2) ,output3 , X, 34, 0, Z)," &
"2 (BC_1, R(3) ,output3 , X, 34, 0, Z)," &
"3 (BC_1, R(4) ,output3 , X, 34, 0, Z)," &
"4 (BC_1, R(5) ,output3 , X, 34, 0, Z)," &
"5 (BC_1, R(6) ,output3 , X, 34, 0, Z)," &
"6 (BC_1, R(7) ,output3 , X, 34, 0, Z)," &
"7 (BC_1, R(8) ,output3 , X, 34, 0, Z)," &
"8 (BC_1, R(1) ,input , X)," &
"9 (BC_1, R(2) ,input , X)," &
"10 (BC_1, R(3) ,input , X)," &
"11 (BC_1, R(4) ,input , X)," &
"12 (BC_1, R(5) ,input , X)," &
"13 (BC_1, R(6) ,input , X)," &
"14 (BC_1, R(7) ,input , X)," &
"15 (BC_1, R(8) ,input , X)," &
"16 (BC_1, B(1) ,output3 , X, 35, 0, Z)," &
"17 (BC_1, B(2) ,output3 , X, 35, 0, Z)," &
"18 (BC_1, B(3) ,output3 , X, 35, 0, Z)," &
"19 (BC_1, B(4) ,output3 , X, 35, 0, Z)," &
"20 (BC_1, B(5) ,output3 , X, 35, 0, Z)," &
"21 (BC_1, B(6) ,output3 , X, 35, 0, Z)," &
"22 (BC_1, B(7) ,output3 , X, 35, 0, Z)," &
"23 (BC_1, B(8) ,output3 , X, 35, 0, Z)," &
"24 (BC_1, B(1) ,input , X)," &
"25 (BC_1, B(2) ,input , X)," &
"26 (BC_1, B(3) ,input , X)," &
"27 (BC_1, B(4) ,input , X)," &
"28 (BC_1, B(5) ,input , X)," &
"29 (BC_1, B(6) ,input , X)," &
"30 (BC_1, B(7) ,input , X)," &
"31 (BC_1, B(8) ,input , X)," &
"32 (BC_1, OE_NEG,input , X)," &
"33 (BC_1, DIR ,input , X)," &
"34 (BC_1, * ,controlr , 0)," &
"35 (BC_1, * ,controlr , 0)" ;

```

5. ábra

Az 5. ábrán az egyes peremfigyelő cellák esetén a lehetséges paraméterek a következők (a 0. peremfigyelő cella példáján tárgyalva):

- a peremfigyelő cella struktúráját megadó, előre definiált azonosító (BC_1),
- a logikai jel meghatározása, amelyre a peremfigyelő cella csatlakozik (A(1)),
- a cella funkciójának megadása (output3), amelynek jelentése: 3 állapotú kimenet,
- azt az értéket adja meg, amelyet az adott peremfigyelő cellába léptetve nem okoz bizonytalan működést a magáramkörben (X). Az „X” jelentése: tetszőleges érték (0/1) beléptethető, nem okoz bizonytalan működést,
- az adott cellához tartozó logikai jelet vezérlő cella sorszáma (34),
- az adott lábhoz tartozó vezérlőjelnek azt az állapotát adja meg, amikor az adott láb kimeneti puffere le van tiltva (0),
- a letiltott kimeneti puffer állapotát adja meg (Z). A „Z” jelentése: a puffer nagyimpedanciás állapotba kerül,
- a 34. és a 35. peremfigyelő cella a 33. „klónja”. A DIR egy bemenet, de két portnál (A és B) vezérlésre is szolgál.

```
end sn74abt8245;
```

6. ábra

Az eszköz leírását a 6. ábrán látható zárósor határolja.

A BSDL-állományokat (fájlokat) az integrált áramkörök gyártói szerkesztik, ellenőrzik és teszik szabadon hozzáférhetővé a felhasználók számára. A BSDL-állományok hiányában a tesztelés automatikus kialakítása nem végezhető el.

kohut.jozsef@kvk.bmf.hu
molnar.zsolt@kvk.bmf.hu